

## Model code file

You can turn the model code file SOEMOLiv1.model into a formatted pdf by running

```
>> latex.publish('SOEMOLiv1.model', 'yourfilename.pdf');
```

### Contents

- Declare model variables and parameters
- Control linearised vs log-linearised variables
- Model equations
- Households
- Capital
- Production
- Local retail
- Exports
- Market clearing
- BoP, UIP and finance
- Monetary policy
- Rest of world
- Definitions & identities
- Carry-on functions
- List of variables

### Declare model variables and parameters

```
11: !variables:transition
12:
13:   C, Y, X, M, N, K, I, A, GDP
14:   P, Pm, Px, W, S, Se, V, Q, Pk, Pk1, Py, T
15:   R, Rstar, Rk, Rk1, U, B
16:   Lambda, F, NGDP, Rl, Ge, G, L
17:   % E, Re, L2E
18:
19:   Pwstar, Rwstar
20:
21: !parameters
22:
23:   alpha, beta, eta, gamman, gammam, delta, nu
24:   a, t, p, f2ngdp, pwstar
25:
26:   chi, xiw, xip, psiy, psix, psii, theta
27:   w, rhor, kappa
28:
29:   rhou, rhob, rhoa, rhot, rhopw
30:
31:   %f0, f1, f2, l2e
32:   g1, g2, g3
33:
34: !shocks:transition
```

```

35:   'Productivity shock' ea, 'Terms of trade shock' et,
36:   'Country risk premium shock' eu, 'Bubble shock' eb
37:   'Policy shock' er
38:   'Foreign inflation shock' epw
39:
40: !shocks:nonlinear
41:   eg, ege

```

## Control linearised vs log-linearised variables

The choice of log-linear variables is critical for unit-root models to have valid first-order approximate solution.

```

49: !variables:log
50:
51:   !allbut
52:   U, A, T, B

```

## Model equations

For some of the equations, we specify their steady-state versions differently from their dynamic versions. This is usually for one of the following two reasons:

- We simplify the dynamic version taking out e.g. adjustment costs (that vanish in the long run) or some of the lags or leads.
- We nail down the model's unit-root processes to get one particular point on the BGP. This is done by assigning arbitrary constants to some of the unit-root variables (such as price levels or productivity).

The steady-state versions follow immediately after the dynamic versions, separated with a double exclamation point, `!!`. The semicolon (to finish the equation) comes only after the steady-state version.

```

71: !equations:transition

```

## Households

Choice variables:  $C_t$ ,  $N_t$ ,  $W_t$ ,  $I_t$ ,  $K_t$ ,  $L_t$ .

Utility function

$$E_0 \sum_t \beta^t \left[ (1 - \chi) \log(C_t - \chi \bar{C}_{t-1}) - \frac{1}{1+\eta} N_t^{1+\eta} \right]$$

Budget constraint

$$L_t = \left[ (1 - \theta)R_{L,t-1} + \theta R_{L,t-1}^* \frac{S_t}{S_{t-1}} \right] L_{t-1} + P_t C_t + P_t I_t (1 + h_{I,t}) - W_t N_t (1 - h_{W,t}) - Q_t \mathcal{K}_t - \Pi_t$$

Wage adjustment costs (private, not social costs)

$$h_{W,t} := \frac{\xi_W}{2} [\log(W_t/\bar{W}_t) - \log(\bar{W}_t/\bar{W}_{t-1})]^2$$

Labour demand curve

$$N_t = (W_t/\bar{W}_t)^{-\nu} \bar{N}_t$$

Investment adjustment costs (private, not social costs)

$$h_{I,t} := \frac{\xi_I}{2} [\log(I_t) - \log(I_{t-1})]^2$$

Law of motion for capital

$$K_t = (1 - \delta)K_{t-1} + I_t$$

Capital services

$$\mathcal{K}_t = K_{t-1}$$

The quantities  $\bar{W}_t$  and  $\bar{N}_t$  are taken as given, and set equal to  $W_t$  and  $N_t$ , respectively, ex post in symmetric equilibrium.

```

114:   (1 - chi)/(C - chi*C{-1}) = Lambda*P !! 1 = Lambda*P*C;
115:
116:   Lambda = beta*Lambda{1}*Rl !! Rl = 1/beta;
117:
118:   xiw/nu*log(dW/dW{-1}) = ...
119:       beta*xiw/nu*log(dW{1}/dW) + (N^eta/(Lambda*W) - (nu-1)/nu) ...
120:       !! nu/(nu-1)*N^eta = Lambda*W;
```

## Capital

We concentrate the equations related to the choice of capital in a separate section.

First, we define the fundamental price of capital, and the fundamental return on capital. Then we introduce a bubble, and the observed market price and return. The observed market variables have a 1 added to their names, Pk1 and Rk1.

Note that the observed market prices and returns are used in the default function below (affecting thus the lending rate) whereas the investment choice is based on the fundamental price and return (see e.g. discussion in Bernanke, Gertler, 1999).

```

138:   K = (1-delta)*K{-1} + I !! delta*K = I;
139:   Pk = P*(1 + psii*(log(I/I{-1}))) - beta*log(I{1}/I)) !! Pk = P;
140:   Rk = (Q + (1-delta)*Pk) / Pk{-1};
141:   Rk{1} = Rl;
142:
143:   B = rhob*B{-1} + eb !! B = 0;
144:   Pk1 = Pk*exp(B) !! Pk1 = Pk;
145:   Rk1 = (Q + (1-delta)*Pk1) / Pk1{-1} !! Rk1 = Rk;

```

## Production

Choice variables:  $Y_t$ ,  $N_t$ ,  $M_t$ ,  $\mathcal{K}_t$ .

Production function

$$Y_t = N_t^{\gamma_N} M_t^{\gamma_M} \mathcal{K}_t^{1-\gamma_N-\gamma_M} \exp(A_t)$$

The value to be maximised

$$E_0 \sum_t \beta \Lambda_t [P_{Y,t} Y_t (1 - h_{Y,t}) - W_t N_t - P_{M,t} M_t - Q_t \mathcal{K}_t]$$

Costs of changing input factor proportions

$$h_{Y,t} := \frac{\xi_Y}{2} [\log(N_t/M_t) - \log(N_{t-1}/M_{t-1})]^2$$

```

166:   Y = exp(A) * K{-1}^(1-gamman-gammam) * N^gamman * M^gammam;
167:
168:   gamman*Py*Y = W*N
169:   *(1 + psii/gamman*(log(N/M)-log(N{-1}/M{-1})) - beta*(log(N{1}/M{1})-log(N/M)))
170:   !! gamman*Py*Y = W*N;
171:
172:   gammam*Py*Y = Pm*M ...
173:   *(1 - psii/gammam*(log(N/M)-log(N{-1}/M{-1})) - beta*(log(N{1}/M{1})-log(N/M))) ...
174:   !! gammam*Py*Y = Pm*M;
175:
176:   (1-gamman-gammam)*Py*Y = Q*K{-1};
177:
178:   A - A{-1} = rhoa*(A{-1} - A{-2}) + (1-rhoa)*ea ...
179:   !! A = a;

```

## Local retail

Choice variables:  $P_t$ ,  $D_t$ .

The value to be maximised

$$E_0 \sum_t \beta \Lambda_t [P_t D_t (1 - h_{P,t}) - P_{Y,t} D_t]$$

Price adjustment costs (private, not social costs)

$$h_{P,t} := \frac{\xi_P}{2} [\log(P_t/P_{t-1}) - \log(\bar{P}_t/\bar{P}_{t-1})]^2$$

Demand curve

$$D_t = (P_t/\bar{P}_t)^{-\nu} \bar{D}_t$$

The quantities  $\bar{P}_t$  and  $\bar{D}_t$  are taken as given, and set equal to  $P_t$  and  $D_t$ , respectively, ex post in symmetric equilibrium.

```

203:    xip/nu*log(dP/dP{-1}) = ...
204:    beta*xip/nu*log(dP{1}/dP) + (Py/P - (nu-1)/nu) ...
205:    !! P = nu/(nu-1)*Py;
206:
207:    Pm = S*Pwstar;
```

## Exports

Choice variables:  $X_t$ .

The value to be maximised

$$E_0 \sum_t \beta \Lambda_t [P_{X,t} X_t (1 - h_{X,t}) - P_{Y,t} X_t]$$

Export adjustment costs (private, not social costs)

$$h_{X,t} := \frac{\xi_X}{2} [\log(X_t) - \log(X_{t-1})]^2$$

```

224:    Px = Py*(1 + psix*(log(X/X{-1}) - beta*log(X{1}/X)))
225:    !! Px = Py;
```

## Market clearing

We substitute out for  $D_t = C_t + I_t$  so that there's only one market clearing condition left.

```
233:    Y = C + X + I;
```

## BoP, UIP and finance

The balance of payments equation derives from the households' budget when the profits and private costs are substituted for.

The interest rates are constructed as follows:

- there's no currency risk, only credit risk;
- whatever the currency of denomination, a country risk premium is added to the world interest rate:

$$R_t = R_{W,t} \exp U_t, \quad R_t^* = R_{W,t}^* \exp U_t$$

- on top of that, the lending rate is adjusted to the expected proportion of non-performing loans,  $E[G_{t+1}]$ . The NPLs are given by an ad-hoc, exogenous function  $g()$  increasing in the ratio of the amount to be repaid to return on capital.

$$R_{L,t} = R_t / E_t[G_{t+1}], \quad R_{L,t}^* = R_t^* / E_t[G_{t+1}]$$

- there are risk-neutral arbitrage conditions holding between any pair of corresponding local-currency and foreign-currency rates, e.g.

$$R_t = R_t^* E_t[S_{t+1}] / S_t$$

```
257:    F = F[-1]*Rstar[-1] - Px*X + Pm*M;
258:
259:    Rstar = Rwstar*exp(U) !! Rstar = Rwstar;
260:    R = Rstar*S[1]/S !! R = Rstar;
261:    U = rhou*U[-1] + eu !! U = 0;
262:    Se = S[1];
263:    V = 1-theta + theta*(S/Se[-1]) !! V = 1;
264:
265:    Rl*Ge = R;
266:    L = F;
267:
268:    'Expected loan performance'
269:    Ge = gfn(g1*Rl*L/(Rk1{1}*Pk1*K),g2,g3) + ege !! Ge = G;
270:
271:    'Actual loan performance' G = gfn( ...
272:    g1*Rl[-1]*L[-1]*V/(Rk1*Pk1[-1]*K[-1]), ...
273:    g2,g3) + eg;
```

## Monetary policy

We combine an exchange rate peg (with weight  $w$ ) and a simple inflation targeting rule (with weight  $1-w$ ). The combined rule is designed so that the policy shock,  $\epsilon_{R,t}$ , always means restrictive policy.

```

282:    % log(S) = log(S{-1}) - er;
283:    % R = rhor*R{-1} + (1-rhor)*(&R + kappa/4*(d4P{3} - 1)) + er;
284:
285:    (1-w)*( R - rhor*R{-1} - (1-rhor)*(&R + kappa/4*(d4P{3} - 1)) ) ...
286:    - w*( log(S) - log(S{-1}) ) - er = 0 ...
287:    !! P = p;

```

## Rest of world

```

291:    Px = exp(T)*S*Pwstar;
292:    T - T{-1} = rhot*(T{-1} - T{-2}) + (1-rhot)*et !! T = t;
293:    dPwstar = rhopw*dPwstar + (1-rhopw)*1 + epw !! Pwstar = pwstar;
294:    Rwstar = &Rwstar;

```

## Definitions & identities

```

298:    NGDP = P*C + P*I + Px*X - Pm*M;
299:
300:    log(GDP) = ...
301:        &P*&C/&NGDP*log(C) ...
302:        + &P*&I/&NGDP*log(I) ...
303:        + &Px*&X/&NGDP*log(X) ...
304:        - &Pm*&M/&NGDP*log(M);
305:
306:    !for
307:        P, W, Pwstar
308:    !do
309:        !variables:transition
310:        d?
311:        !equations:transition
312:        d? = ? / ?{-1} !! d? = 1;
313:    !end
314:
315:    !for
316:        P
317:    !do
318:        !variables:transition
319:        d4?
320:        !equations:transition
321:        d4? = ? / ?{-4} !! d4? = 1;
322:    !end

```

## Carry-on functions

The carry-on functions are saved as separate m-files in the working directory at the time the model code is read in by the `model` function.

This is the function used to compute the proportion of NPLs. Each NPL has a recovery rate of  $1 - \mu$ .

```
333: !function y = gfn(x,s,mu)
334:   y = 1 - mu.*logncdf(x,-(s.^2)/2,s);
335: !end
```

## List of variables

in alphabetical order

$A$	Productivity
$B$	Capital price bubble
$C$	Consumption
$D$	Domestic demand
$F$	Foreign debt
$G$	Proportion of non-performing loans
$GDP$	Value added
$I$	Investment
$K$	Capital
$\mathcal{K}$	Capital services
$L$	Loans
$M$	Imports
$P$	Final prices (consumption, investment)
$P_K$	Fundamental price of capital
$P_K^1$	Observed market price of capital
$P_M$	Import prices
$P_W^*$	World prices (in foreign currency)
$P_X$	Export prices
$P_Y$	Marginal cost of gross output
$Q$	Rental price of capital
$R$	Wholesale local-currency rate
$R^*$	Wholesale foreign-currency rate
$R_W^*$	World foreign-currency risk-free rate
$R_K$	Fundamental return on capital
$R_K^1$	Observed market return on capital
$R_L$	Local-currency lending rate
$R_L^*$	Foreign-currency lending rate
$S$	Nominal exchange rate
$T$	Terms of trade
$U$	Country risk premium
$V$	Ex-post valuation effect
$W$	Nominal wage rate
$X$	Export
$Y$	Gross output